DECUS
PROGRAM LIBRARY

DECUS NO.	FOCAL8-248
TITLE	FOCTXT - TEXT INPUT-OUTPUT PATCH TO FOCAL-1969
AUTHOR	F. R. Johnson
COMPANY	Dow Badische Company Freeport, Texas
DATE	November 3, 1972
SOURCE LANGUAGE	PAL III

FOCAL was developed to be used as a problem solving language. As such input to a user program is restricted to numeric entries.

The following patch will allow two new functions, FRSC(X) (Read String of Characters) and FTSC(X) (Type String of Characters). These functions allow the user of FOCAL to input and output text that is not included in the body of the user program. Incoming text is delimited by a carriage return. The carriage return is not stored in the text buffer.

To allow incoming text to be read and stored the user program executes S X=FRSC(0) where X is a dummy variable and 0 means to start storing in the first word of the buffer. Asuume that the user now enters - there will be no echo - the following:

ABCDEFGHIJKLM CR

Now if S X=FTSC(0) is executed, ABCDEFGHIJKLM will type out. If S X=FTSC(5) is executed, them KLM will type out. The user must keep up with where he is in the buffer when storing more than one text string.

For example, if the above text was stored starting at 0, the next possible starting point is given by N=FITR(L/2)+2. In this case N=8.

Leader-trailer and rubouts inbedded in text are ignored. Text is trimmed and stored two characters per core word. Text input is delimited by carriage return.

The text buffer will hold approximately 120 (10) characters depending on blocking.

These routines use pointers and hash code areas for FNEW and FCOM.

```
                / TEXT INPUT OUTPUT
                / PATCH TO FOCAL 1969
                / PRJ     10/30/72

                    EXIT=JMP I 136
                    START=4400    / 5400 STRIPPED


0035  4400          *35;XRSC      / LAST AVAILABLE CORE

0410  4400          *410;XRSC     / POINTER

0411  4425          *411;XTSC     / POINTER

2201  2661          *2201;2661    / RSC HASH CODE

2202  2671          *2202;2671    / TSC HASH CODE

                    *START

                    READ=JMS XRBX
                    TYPE=JMS TLSX

4400  4453  XRSC, JMS I 53    / GET ARGUEMENT
4401  1246      TAD BASE
4402  3250      DCA POINT
4403  3044      DCA 44          / CLEAR FLAG
4404  4251  NEXTB, READ        / GET FIRST ALPHA
4405  7106      CLL RTL
4406  7036      RTL
4407  7006      RTL             / SHIFT UP
4410  3247      DCA TEMP
4411  4251      READ            / GET 2ND ALPHA
4412  1247      TAD TEMP
4413  3650      DCA I POINT     / STORE 2 IN 1
4414  2250      ISZ POINT
4415  3247      DCA TEMP
4416  5204      JMP NEXTB
4417  7200  ENDB, CLA
4420  1247      TAD TEMP
4421  3650      DCA I POINT
4422  2250      ISZ POINT
4423  3650      DCA I POINT     / 00 OR 0000 AT END
4424  5536      EXIT

4425  4453  XTSC, JMS I 53    / GET ARGUEMENT
4426  1246      TAD BASE
4427  3250      DCA POINT
4430  3044      DCA 44          / CLEAR FLAG
4431  1650  NEXTT, TAD I POINT
4432  7450      SNA
4433  5536      EXIT
4434  3247      DCA TEMP
4435  2250      ISZ POINT
4436  1247      TAD TEMP
4437  7112      CLL RTR
```

```
4440  7010      RTR
4441  7010      RTR             / SHIFT DOWN
4442  4272      TYPE
4443  1247      TAD TEMP
4444  4272      TYPE
4445  5231      JMP NEXTT
4446  4507  BASE,BUFFER
4447  0000  TEMP,0
4450  0000  POINT,0

4451  0000  KRBX,0
4452  4464      JMS I 64        / FOCAL INPUT CTRL/C BREAKOUT
4453  1267      TAD MRUB
4454  7450      SNA
4455  5252      JMP KRBX+1      / IGNORE RUBOUTS
4456  1270      TAD MCR
4457  7450      SNA
4460  5217      JMP ENDR        / CR DELIMITS TEXT
4461  1271      TAD PCR
4462  0266      AND MK77
4463  7450      SNA
4464  5252      JMP KRBX+1      / IGNORE L-T
4465  5651      JMP I KRBX
4466  0077  MK77,77
4467  7401  MRUB,0-377
4470  3162  MCR,377-215
4471  0215  PCR,215

4472  0000  TLSX,0
4473  0266      AND MK77
4474  7450      SNA
4475  5536      EXIT
4476  1304      TAD M40
4477  7510      SPA
4500  1335      TAD K100
4501  1336      TAD K240
4502  4463      JMS I 63        / FOCAL OUTPUT  ALSO CLEARS AC
4503  5672      JMP I TLSX
4504  7740  M40,0-40
4505  0100  K100,100
4506  0240  K240,240

4507  0000  BUFFER,0

BASE    4446
BUFFER  4507
ENDR    4417
EXIT    5536
KRBX    4451
K100    4505
K240    4506
MCR     4470
MK77    4466
MRUB    4467
M40     4504
NEXTR   4404
NEXTT   4431
PCR     4471
POINT   4450
READ    4251
START   4400
TEMP    4447
TLSX    4472
TYPE    4272
XRSC    4400
XTSC    4425
```